

**BASIC/EXTENDED BASIC  
CHARACTER OFFSET  
DEMONSTRATION**

The following two programs demonstrate the HEX 60 character offset when printing text to the screen from the BASIC or Extended BASIC environment. The first program is to be used with Extended BASIC. When assembling the program, name the object file XBASD and remember, do not use 'C' as one of the options. Also, use the following program from Extended BASIC to load and link the Assembly program.

```
100 CALL INIT
110 CALL LOAD("DSK1.XBASD")
120 CALL CLEAR
130 LINPUT "ENTER STRING: ":A$
140 CALL CLEAR
150 CALL LINK("START",A$)
```

Because Extended BASIC utilizes the LINPUT command, all punctuation marks are allowed in the string.

The next program is a slight modification of the first and is for use with TI BASIC programs when the Editor/Assembler Command Module is inserted. Again, when assembling, name the object file XBASD. This time, use of the 'C' option is recommended. Also, include the BASIC Support file (BSCSUP) from the E/A PART A diskette on the diskette containing the file XBASD. Use the following program to load and link the Assembly program.

```
100 CALL INIT
110 CALL LOAD("DSK1.XBASD","DSK1.BSCSUP")
120 CALL CLEAR
130 INPUT "ENTER STRING: ":A$
140 CALL CLEAR
150 CALL LINK("START",A$)
```

TI BASIC does not contain the LINPUT command, therefore, commas are not allowed.

This material is given to you by Texas Instruments Incorporated without representation or warranty of any kind. Therefore, we assume no responsibility and shall have no liability, consequential or otherwise, of any kind arising from its use. This material was developed by and is considered to be the property of Texas Instruments. We therefore reserve the right to use, publish, reproduce, or sell this material in any manner desired with out compensation of any kind.

Example: Say your collection of baseball cards consists of 20 cards broken down into the following categories.

<u>PLAYER</u>	<u>NUMBER OF CARDS</u>	<u>TEAM</u>
A) Reggie Jackson	4	Yankees
B) Pete Rose	2	Phillies
C) Nolan Ryan	2	Astros
D) Steve Garvey	5	Dodgers
E) Carlton Fisk	3	Red Sox
F) Al Oliver	1	Rangers
G) George Brett	1	Royals
H) Willie Stargell	1	Pirates
I) Thurmon Munson	1	Yankees

The enclosed program will allow one to enter the players name, number of cards, and the players' team name. The program will then alphabetize the players, record the number of cards, and cross reference the team.

### User Instructions

- A) Enter the following program and RUN
- B) To enter new cards press A
  - 1) Computer will prompt for the number of players. Since our example has 9 players, press 9 then ENTER.
  - 2) Computer will display format in which to enter player information.
  - 3) Enter players name in the following format "Jackson, Reggie", 4, Yankees.
  - 4) Press ENTER.
  - 5) Computer will prompt the user until all 9 players have been entered.
  - 6) After all 9 players have been entered computer will return to main index.
- C) To delete cards press B.
  - 1) Computer will prompt for the proper transaction.
  - 2) To delete players name, press 1 then ENTER.
  - 3) Computer will prompt for the players' name.
  - 4) Enter players name in the following format: "Jackson, Reggie".
  - 5) Press ENTER.
  - 6) Computer will prompt for another player. Enter Y for yes or N for no. If yes computer will prompt for players' name, if no computer returns to main index.
  - 7) To delete a certain number of cards press 2 then ENTER.
  - 8) Computer will prompt for the number of cards to be deleted. Press that number and then ENTER.
  - 9) Computer will then prompt in the format described in parts 3-6 of this section.
  - 10) To return to main index press 3 then ENTER.
- D) To reference players' name press C.
  - 1) Computer will prompt in the format described in parts 3-5 of section C.
  - 2) Computer will display players' name, number of cards, and team name. If players' name is not found, computer will prompt for another player.
- E) To reference a team press D.
  - 1) Computer will prompt for the name of the team.
  - 2) Enter YANKEES as the team name and press ENTER.
  - 3) Computer will display all players' names, number of cards, and the team name
  - 4) Computer will then prompt for another team.
- F) To read Data (CS1) press E.
  - 1) Follow instructions that computer prompts for.
- G) To record Data (CS1) press F.
  - 1) Follow instructions that computer prompts for.

- H) To list all players and teams press G.
  - 1) Computer will list, alphabetically, five players at a time.
  - 2) Press C to continue to list teams.
  - 3) Press Q to return to main index.
- I) To leave program press H.

This material is given to you by Texas Instruments Incorporated without representation or warranty of any kind. Therefore, we assume no responsibility and shall have no liability, consequential or otherwise, of any kind arising from its use. This material was developed by and is considered to be the property of Texas Instruments. We therefore reserve the right to use, publish, reproduce, or sell this material in any manner desired without compensation of any kind.

```

100 CALL CLEAR
110 DIM A$(100),B(100),C$(100)
120 PRINT "PRESS THE CORRESPONDING"
130 PRINT "KEY FOR THE APPROPRIATE"
140 PRINT "FUNCTION"
150 PRINT
160 PRINT
170 PRINT "PRESS FOR"
180 PRINT
190 PRINT "A-----ADDING CARDS"
200 PRINT "B-----DELETING CARDS"
210 PRINT "C-----TO REFERENCE PLAYERS"
220 PRINT "D-----TO REFERENCE TEAMS"
230 PRINT "E-----TO READ DATA"
240 PRINT "F-----TO RECORD DATA"
250 PRINT "G-----TO LIST ALL TEAMS"
260 PRINT "H-----TO QUIT"
270 CALL KEY(0,KEY,STATUS)
280 IF STATUS=0 THEN 270
290 IF (KEY<65)+(KEY>72) <=-1 THEN 270
300 KEY=KEY-64
310 CALL CLEAR
320 ON KEY GOTO 480,940,1180,1270,340,420,1380,330
330 END
340 OPEN #1:"CS1",INPUT,INTERNAL,FIXED 192
350 INPUT #1:CT
360 FOR I=1 TO CT
370 INPUT #1:A$(I),B(I),C$(I)
380 NEXT I
390 CLOSE #1
400 CALL CLEAR
410 GOTO 120
420 OPEN #1:"CS1",OUTPUT,INTERNAL,FIXED 192
430 PRINT #1:CT
440 FOR I=1 TO CT
450 PRINT #1:A$(I),B(I),C$(I)
460 NEXT I
470 GOTO 390
480 INPUT "NUMBER OF PLAYERS":CD
490 CALL CLEAR
500 GOTO 550
510 PRINT "INPUT THE INFORMATION"

```

```

520 PRINT "IN THE FOLLOWING FORM AT"
530 PRINT
540 RETURN
550 GOSUB 510
560 PRINT ""LASTNAME,FIRSTNAME"";";";"# OF CARDS";";";"TEAM NAME"
570 FOR I=1 TO CD
580 PRINT
590 PRINT
600 PRINT "NAME,#OF CARDS,TEAM"
610 INPUT AA$,BB,CC$
620 CT=CT+1
630 FOR J=1 TO CT
640 IF A$(J)="" THEN 770
650 IF AA$<=A$(J) THEN 670
660 NEXT J
670 IF AA$=A$(J) THEN 680 ELSE 720
680 B(J)=B(J)+BB
690 CT=CT-1
700 NEXT I
710 GOTO 400
720 FOR K=CT TO J STEP -1
730 A$(K)=A$(K-1)
740 B(K)=B(K-1)
750 C$(K)=C$(K-1)
760 NEXT K
770 A$(J)=AA$
780 B(J)=BB
790 C$(J)=CC$
800 GOTO 700
810 GOSUB 510
820 PRINT ""LASTNAME,FIRSTNAME""
830 INPUT AA$
840 FOR I=1 TO CT
850 IF AA$=A$(I) THEN 880
860 NEXT I
870 GOTO 1200
880 CALL CLEAR
890 PRINT A$(I);B(I);C$(I)
900 PRINT
910 IF FLAG=0 THEN 940
920 RETURN
930 PRINT
940 PRINT "1---TO DELETE PLAYERS NAME"
950 PRINT "2---TO DELETE # OF CARDS"
960 PRINT "3---TO RETURN TO MAIN INDEX"
970 PRINT
980 INPUT "WHICH TRANSACTION? ";TR

```

```

990 IF (TR<1)+(TR>3)<=-1 THEN 98
0
1000 CALL CLEAR
1010 ON TR GOTO 1020,1120,400
1020 FLAG=1
1030 GOSUB 810
1040 FOR K=1 TO CT
1050 A$(K)=A$(K+1)
1060 B(K)=B(K+1)
1070 C$(K)=C$(K+1)
1080 NEXT K
1090 CT=CT-1
1100 FLAG=0
1110 GOTO 900
1120 INPUT "# OF CARDS TO BE DEL
ETED ":DE
1130 FLAG=1
1140 GOSUB 810
1150 FLAG=0
1160 B(I)=B(I)-DE
1170 GOTO 890
1180 FLAG=1
1190 GOSUB 810
1200 PRINT
1210 INPUT "ANOTHER PLAYER? (Y/N)
":PL$
1220 IF PL$="Y" THEN 1250
1230 FLAG=0
1240 GOTO 400
1250 GOSUB 820
1260 GOTO 1200
1270 FLAG=1
1280 INPUT "ENTER TEAM NAME ":TE
$
1290 FOR I=1 TO CT
1300 IF TE$=C$(I) THEN 1360
1310 NEXT I

```

```

1320 INPUT "ANOTHER TEAM? (Y/N)
":PL$
1330 IF PL$="Y" THEN 1280
1340 FLAG=0
1350 GOTO 400
1360 GOSUB 890
1370 GOTO 1310
1380 FLAG=1
1390 FOR I=1 TO CT
1400 IF I-X=5 THEN 1510
1410 GOSUB 890
1420 NEXT I
1430 FLAG=0
1440 X=0
1450 PRINT "PRESS Q TO RETURN"
1460 CALL KEY(0,KEY,STATUS)
1470 IF STATUS=0 THEN 1460
1480 IF KEY=67 THEN 1420
1490 IF KEY=81 THEN 400
1500 GOTO 1460
1510 GOSUB 890
1520 X=X+5
1530 PRINT "PRESS C TO CONTINUE"
1540 GOTO 1460

```

This material is given to you by Texas Instruments Incorporated without representation or warranty of any kind. Therefore, we assume no responsibility and shall have no liability, consequential or otherwise, of any kind arising from its use. This material was developed by and is considered to be the property of Texas Instruments. We therefore reserve the right to use, publish, reproduce, or sell this material in any manner desired with out compensation of any kind.

Texas Instruments Incorporated, 1977

\*SCREEN DUMP\*  
\*PROGRAM\*

THE FOLLOWING PROGRAM RANDOMLY PLACES CHARACTERS ON THE SCREEN AND PRINTS THEM TO AN RS232 COMPATIBLE PRINTER. THIS PROGRAM RUNS IN BOTH TI BASIC AND TI EXTENDED BASIC. THIS PROGRAM WILL NOT TRANSFER USER DEFINED CHARACTERS FROM THE SCREEN TO THE PRINTER.

```
100 REM ~~~~~
110 REM SCREEN DUMP TO
120 REM PRINTER ROUTINE
130 REM ~~~~~
140 CALL CLEAR
150 OPEN #1:"RS232",VARIABLE 32
160 REM ~~~~~
170 REM BORDER AROUND SCREEN
180 REM ~~~~~
190 CALL HCHAR(1,1,42,32),
200 CALL HCHAR(24,1,42,32)
210 CALL VCHAR(2,1,42,22)
220 CALL VCHAR(2,32,42,22)
230 REM ~~~~~
240 REM PRINT CHARACTERS
250 REM RANDOMLY TO SCREEN
260 REM ~~~~~
270 FOR A=1 TO 50
280 RANDOMIZE
290 ROW=INT(22*RND)+2
300 COLUMN=INT(30*RND)+2
310 CALL HCHAR(ROW,COLUMN,42)
320 NEXT A
330 REM ~~~~~
340 REM READ SCREEN AND
350 REM PRINT CHARACTERS TO
360 REM PRINTER
370 REM ~~~~~
380 FOR A=1 TO 24
390 FOR B=1 TO 32
400 CALL GCHAR(A,B,X)
410 PRINT #1:CHR$(X);
420 NEXT B
430 NEXT A
440 END
```

THIS MATERIAL IS GIVEN TO YOU BY TEXAS INSTRUMENTS INCORPORATED WITHOUT REPRESENTATION OR WARRANTY OF ANY KIND. THEREFORE, WE ASSUME NO RESPONSIBILITY AND SHALL HAVE NO LIABILITY, CONSEQUENTIAL OR OTHERWISE, OF ANY KIND ARISING FROM ITS USE. THIS MATERIAL WAS DEVELOPED BY AND IS CONSIDERED TO BE THE PROPERTY OF TEXAS INSTRUMENTS. WE THEREFORE RESERVE THE RIGHT TO USE, PUBLISH, REPRODUCE, OR SELL THIS MATERIAL IN ANY MANNER DESIRED WITHOUT COMPENSATION OF ANY KIND.

# TEXAS INSTRUMENTS



Dear Customer:

Thank you for contacting Texas Instruments regarding programming your TI-99/4A home computer. This program has been developed to format and round a number entered from the keyboard. The number will be in dollar format with zeros displayed and will round any amount over one-half a cent to a full cent. The program lists as follows:

```
100 INPUT "Type in a number. ":X
110 X=INT((X+.005)*100)
120 IF LEN(STR$(X))=1 THEN 240
130 W$=SEG$(STR$(X),LEN(STR$(X)),1)
140 T$=SEG$(STR$(X),LEN(STR$(X))-1,2)
150 X=X/100
160 IF VAL(T$)=0 THEN 200
170 IF VAL(W$)=0 THEN 220
180 PRINT "$";X
190 GOTO 250
200 PRINT " $ ";STR$(X)&".00"
210 GOTO 250
220 PRINT " $ ";STR$(X)&"0"
230 GOTO 250
240 PRINT " $ 0.0"&STR$(X)
250 PRINT "Is the rounded $ amount.":::
260 GOTO 100
```

The program was developed to aid in number formatting and is given to you by Texas Instruments Incorporated without representation of warranty of any kind. Therefore, we assume no responsibility and shall have no liability, consequential or otherwise, of any kind arising from its use. This program was developed by and is considered to be the property of Texas Instruments. We therefore reserve the right to use, publish, reproduce or sell the material in any manner desired.

I hope this information will help you better utilize the BASIC language.

Yours truly,

Consumer Relations

FE-06-CR-005/07

\*GRAPHIC MOVEMENT PROGRAM\*

THE FOLLOWING PROGRAM UTILIZES BOTH THE KEYBOARD AS WELL AS JOYSTICKS TO INCORPORATE GRAPHIC MOVEMENT WITH THE 99/4A HOME COMPUTER. IF ARE NOT USING JOYSTICKS, PRESSING THE "Q" KEY WILL INACT THE FIRE PROCEDURE.

TI DOES NOT WARRANT THAT THIS PROGRAM WILL BE FREE FROM ERROR OR WILL MEET THE SPECIFIC REQUIREMENTS OF THE USER. THE USER ASSUMES COMPLETE RESPONSIBILITY FOR ANY DECISION MADE OR ACTIONS TAKEN BASED ON INFORMATION OBTAINED USING THIS PROGRAM.

```
100 REM GRAPHIC MOVEMENT PROGRAM
110 CALL SCREEN(5)
120 CALL COLOR(1,12,1)
130 CALL CLEAR
140 CALL CHAR(33,"0018183C3C7E7EFF")
150 CALL CHAR(34,"01071F7F7F1F0701")
160 CALL CHAR(35,"FFFFFFFFFFFFFF")
170 CALL CHAR(36,"80E0FBFEFEFB8E080")
180 CALL CHAR(37,"FF7E7E3C3C181800")
190 CALL CHAR(133,"0000000000000000")
200 REM INITIAL CHARACTER PLACEMENT
210 R=11
220 C=15
230 CALL HCHAR(R,C,33)
240 CALL HCHAR(R+1,C-1,34)
250 CALL HCHAR(R+1,C,35)
260 CALL HCHAR(R+1,C+1,36)
270 CALL HCHAR(R+2,C,37)
280 CALL KEY(0,KEY,STATUS)
290 REM TEST FOR FIRE BUTTON
300 CALL KEY(1,FIRE,STATUS)
310 IF FIRE<>18 THEN 420
320 SWITCH=SWITCH*-1
330 CALL SCREEN(12)
340 CALL COLOR(1,5,1)
350 REM EXPLOSION
360 CALL SOUND(-200,-6,2)
370 CALL SOUND(-500,-7,4)
380 IF SWITCH=-1 THEN 420
390 CALL SCREEN(5)
400 CALL COLOR(1,12,1)
410 REM TEST FOR KEYBOARD INPUT
420 IF KEY=69 THEN 610
430 IF KEY=88 THEN 640
440 IF KEY=83 THEN 670
450 IF KEY=88 THEN 700
460 IF KEY=101 THEN 610
470 IF KEY=100 THEN 640
480 IF KEY=115 THEN 670
490 IF KEY=120 THEN 700
500 REM TEST FOR JOYSTICK INPUT
510 CALL JOYST(1,X,Y)
520 IF X=0 THEN 530 ELSE 560
530 IF Y=0 THEN 230
540 IF Y=4 THEN 610
550 IF Y=-4 THEN 700
560 IF X=4 THEN 570 ELSE 580
570 IF Y=0 THEN 640
580 IF X=-4 THEN 590
590 IF Y=0 THEN 670 ELSE 280
600 REM MOVEMENT SUBROUTINES
610 GOSUB 740
620 R=R-1
630 GOTO 230
640 GOSUB 740
650 C=C+1
660 GOTO 230
670 GOSUB 740
680 C=C-1
690 GOTO 230
700 GOSUB 740
710 R=R+1
720 GOTO 230
730 REM REPLACES PREVIOUS LOCATION WITH BLANK SPACE
740 CALL HCHAR(R,C,133)
750 CALL HCHAR(R+1,C-1,133)
760 CALL HCHAR(R+1,C,133)
770 CALL HCHAR(R+1,C+1,133)
780 CALL HCHAR(R+2,C,133)
790 RETURN
```

# HEX-CODE DEFINITION

33

								00
			X	X				18
			X	X				18
		X	X	X	X			3C
		X	X	X	X			3C
	X	X	X	X	X	X		7E
	X	X	X	X	X	X		7E
X	X	X	X	X	X	X	X	FF

34

							X	01
				X	X	X	X	07
		X	X	X	X	X	X	1F
X	X	X	X	X	X	X	X	7E
X	X	X	X	X	X	X	X	7F
		X	X	X	X	X	X	1F
			X	X	X	X	X	07
							X	01

35

X	X	X	X	X	X	X	X	FF
X	X	X	X	X	X	X	X	FF
X	X	X	X	X	X	X	X	FF
X	X	X	X	X	X	X	X	FF
X	X	X	X	X	X	X	X	FF
X	X	X	X	X	X	X	X	FF
X	X	X	X	X	X	X	X	FF
X	X	X	X	X	X	X	X	FF
X	X	X	X	X	X	X	X	FF

36

X								80
X	X	X						E0
X	X	X	X	X				F8
X	X	X	X	X	X	X	X	FE
X	X	X	X	X	X	X	X	FE
X	X	X	X	X	X	X	X	F8
X	X	X	X	X	X	X	X	E0
X	X	X						80

37

X	X	X	X	X	X	X	X	FF
	X	X	X	X	X	X	X	7E
	X	X	X	X	X	X	X	7E
		X	X	X	X			3C
		X	X	X	X			3C
			X	X				18
			X	X				18
								00

CHARACTER #

HEX-CODE

33

0018183C3C7E7EFF

34

01071F7F7F1F0701

35

FFFFFFFFFFFFFFFF

36

80E0F8FEFEF8E080

37

FF7E7E3C3C181800

NOTE: REFER TO USER REFERENCE MANUAL - PAGE II-77.

SAMPLE PROGRAM FOR COINCIDENCE

EXTENDED BASIC REQUIRED

100 CALL CLEAR	CLEAR SCREEN.
110 CALL SCREEN(5)	CHANGE SCREEN COLOR TO DARK BLUE.
120 CALL CHAR(96,"0001031F3F 77FFAF0F5FFF1F00000000808080 80C1E3F6FCFEFBF1C0")	DEFINE CHARACTER FOR DOUBLE SIZE AND MAGNIFIED SPRITE. THIS IS FOR SPRITE #1.
130 CALL CHAR(100,"000000008 4CE5D7E7CCF86000000000000000 0000000000000000")	DEFINE CHARACTER FOR MAGNIFIED SPRITE. THE ZEROS LINE UP SPRITE #2 WITH THE MOUTH OF SPRITE #1.
140 CALL MAGNIFY(4)	MAKE SPRITES DOUBLE SIZE AND MAGNIFIED.
150 CALL SPRITE(#1,96,9,100, 256)	SPRITE #1 IS MEDIUM RED AND STARTS AT THE RIGHT SIDE OF THE SCREEN.
160 CALL SPRITE(#2,100,12,10 0,1)	SPRITE #2 IS LIGHT YELLOW AND STARTS AT THE LEFT SIDE OF THE SCREEN.
170 RANDOMIZE	RESET RANDOM NUMBER GENERATOR.
180 Y=INT(RND*9)+1	RANDOM VERTICAL SPEED FOR SPRITE #1.
190 Z=INT(RND*9)+1	RANDOM VERTICAL SPEED FOR SPRITE #2.
200 CALL MOTION(#1,Y,-20)	PLACE SPRITE #1 INTO MOTION.
210 CALL MOTION(#2,-Z,25)	PLACE SPRITE #2 INTO MOTION.
220 CALL COINC(#1,#2,7,A)	CHECK FOR COINCIDENCE BETWEEN #1 AND #2.
230 IF NOT A THEN 220	IF NO COINCIDENCE, CHECK AGAIN (LINE 220).
240 CALL DELSPRITE(#2)	MAKE SPRITE #2 DISAPPEAR.
250 FOR I=6 TO 16	SET UP LOOP FOR COLOR AND SOUND.
260 CALL COLOR(#1,I)	CHANGE COLOR OF #1 TO COLOR CODE OF I.
270 CALL SOUND(10,100*I,1)	CREATE SOUND THAT GOES UP IN PITCH.
280 NEXT I	INCREMENT I UNTIL IT IS GREATER THEN 16.
290 CALL COLOR(#1,9)	RESET THE COLOR OF SPRITE #1 TO MEDIUM RED.
300 GOTO 160	GOTO 160 AND PUT SPRITE #2 BACK ON SCREEN.
310 END	END OF PROGRAM.

TO STOP PROGRAM PRESS FCTN 4 (CLEAR).

TI DOES NOT WARRANT THAT THIS PROGRAM WILL BE FREE FROM ERROR OR WILL MEET THE SPECIFIC REQUIREMENTS OF THE USER. THE USER ASSUMES COMPLETE RESPONSIBILITY FOR AND DECISIONS MADE OR ACTIONS TAKEN BASED ON INFORMATION OBTAINED USING THIS PROGRAM.

## DERIVED FUNCTIONS

The following functions, which are not implemented in TI-BASIC, can be utilized with the formulae below:

$$\text{COT}(X) = 1/\text{TAN}(X)$$

$$\text{CSC}(X) = 1/\text{SIN}(X)$$

$$\text{SEC}(X) = 1/\text{COS}(X)$$

$$\text{SINH}(X) = (\text{EXP}(X) - \text{EXP}(-X))/2$$

$$\text{COSH}(X) = (\text{EXP}(X) + \text{EXP}(-X))/2$$

$$\text{TANH}(X) = \text{SINH}(X)/\text{COSH}(X)$$

$$\text{CSCH}(X) = 1/\text{SINH}(X)$$

$$\text{SECH}(X) = 1/\text{COSH}(X)$$

$$\text{COTH}(X) = 1/\text{TANH}(X)$$

$$\text{SIN}^{-1}(X) = \text{ATN}(X/\text{SQR}(1-X*X))$$

$$\text{COS}^{-1}(X) = \text{PI}/2 - \text{ATN}(X/\text{SQR}(1-X*X))$$

$$\text{COT}^{-1}(X) = \text{PI}/2 - \text{ATN}(X)$$

$$\text{CSC}^{-1}(1/\text{SQR}(X*X-1)) + (\text{SGN}(X) - 1) * \text{PI}$$

$$\text{SEC}^{-1} = \text{ATN}(\text{SQR}(X*X-1)) - (\text{SGN}(X) - 1) * \text{PI}$$

$$\text{SINH}^{-1}(X) = \text{LOG}(X + \text{SQR}(X*X+1))$$

$$\text{COSH}^{-1}(X) = \text{LOG}(X + \text{SQR}(X*X-1)), \text{ for } X \geq 1 \text{ only}$$

$$\text{TANH}^{-1}(X) = \text{LOG}((1+X)/(1-X))/2, \text{ for } X^2 < 1 \text{ only}$$

$$\text{SECH}^{-1} = \text{LOG}((1 + \text{SQR}(1-X*X))/X), \text{ for } 0 < X \leq 1 \text{ only}$$

$$\text{COTH}^{-1} = \text{LOG}((X+1)/(X-1))/2, \text{ for } X^2 > 1 \text{ only}$$

$$\text{CSCH}^{-1} = \text{LOG}((\text{SGN}(X) + \text{SQR}(X*X+1)+1)/X)$$

$$\text{PI} = 3.14159265359$$

\*DISPLAY-AT/ACCEPT-AT\*  
\*PROGRAM IN TI BASIC\*

THE FOLLOWING PROGRAM IS WRITTEN TO DEMONSTRATE THE CAPABILITY IN TI BASIC TO DISPLAY INFORMATION AT A GIVEN SCREEN LOCATION AND ACCEPT KEYBOARD INPUT AT ANY LOCATION ON THE SCREEN.

```
100 CALL CLEAR
110 QUESTION$="WHAT IS YOUR NAME?"
120 B=LEN(QUESTION$)
130 C=INT(16-(B/2))
140 REM ~~~~~
150 REM ROUTINE TO DISPLAY
160 REM QUESTION ON SCREEN
170 REM ~~~~~
180 FOR COUNT=1 TO B
190 MSG$=SEG$(QUESTION$,COUNT,1)
200 CALL HCHAR(6,C,ASC(MSG$))
210 C=C+1
220 NEXT COUNT
230 REM ~~~~~
240 REM ROUTINE TO SCAN
250 REM KEYBOARD FOR
260 REM RESPONSE
270 REM ~~~~~
280 COLUMN=11
290 FOR COUNT=1 TO 10
300 CALL KEY(3,KEY,STATUS)
310 IF STATUS<>1 THEN 300
320 CALL HCHAR(8,COLUMN,KEY)
330 IF KEY=13 THEN 420
340 NAME$=NAME$&CHR$(KEY)
350 COLUMN=COLUMN+1
360 NEXT COUNT
370 REM ~~~~~
380 REM ROUTINE TO COMPILE
390 REM KEYBOARD RESPONSE
400 REM INTO A STRING
410 REM ~~~~~
420 GREETING$="HELLO "&NAME$
430 B=LEN(GREETING$)
440 C=INT(16-(B/2))
450 FOR COUNT=1 TO B
460 MSG$=SEG$(GREETING$,COUNT,1)
470 CALL HCHAR(10,C,ASC(MSG$))
480 C=C+1
490 NEXT COUNT
500 END
```

THIS MATERIAL IS GIVEN TO YOU BY TEXAS INSTRUMENTS INCORPORATED WITHOUT REPRESENTATION OR WARRANTY OF ANY KIND. THEREFORE, WE ASSUME NO RESPONSIBILITY AND SHALL HAVE NO LIABILITY, CONSEQUENTIAL OR OTHERWISE, OF ANY KIND ARISING FROM ITS USE. THIS MATERIAL WAS DEVELOPED BY AND IS CONSIDERED TO BE THE PROPERTY OF TEXAS INSTRUMENTS. WE THEREFORE RESERVE THE RIGHT TO USE, PUBLISH, REPRODUCE, OR SELL THIS MATERIAL IN ANY MANNER DESIRED WITHOUT COMPENSATION OF ANY KIND.